
• Zusammenfassung

Die (lineare) Diskriminanzanalyse hat zum Ziel, anhand von Objekten, an denen 
mehrere Merkmale gemessen wurden, die aber a-priori einer bestimmten Klasse 
angehören, lineare Grenzen zwischen diesen zwei oder mehr Klassen zu ziehen, 
um neue Objekte einer dieser Klassen zuzuordnen. 

SAS bietet mit der Prozedur discrim ein Werkzeug, diese Trennung(n) zu 
berechnen. Doch leider gibt es keine Grafik, die die lineare(n) Trennfunktion(en) 
zeichnet. Oder aber, ich kenne sie nicht.

Anhand der im Output zu findenden Daten ist es jedoch möglich, die Parameter 
der Funktion(en) zu errechnen und dann mittels sgplot auch grafisch in der 
Ebene darzustellen. Darüber hinaus kann anhand der ODS-Ergebnistabellen 
ermittelt werden, welche a-priori Objekte falsch klassifiziert würden. Dieses kann 
mittels  Kodierung ebenso in der sgplot-Grafik verdeutlicht werden.

Als Nebeneffekt ist es dann ebenso möglich, die Wirkung der verwendeten 
a-priori Wahrscheinlichkeiten auf die Lage der Trennfunktionen anschaulich 
darzustellen.

SAS® kann, 

aber wie?

(a1+b1*y)/c1 = (a2+*b2*y)/c2

a1=2*&c12.
b1=(&p12.-&p22.)
c1=(&p11.-&p21.)

a2=2*&c23.
b2=(&p22.-&p32.)
c2=(&p21.-&p31.)

c2*(a1+b1*y) = c1*(a2+*b2*y)

c2*a1 - c1*a2  = c1*b2*y -c2*b1*y

c2*a1 - c1*a2  = y * (c1*b2 -c2*b1)

y = (c2*a1 - c1*a2)/(c1*b2 -c2*b1)

zero = ((&p21.-&p31.)*2*&c12. - (&p11.-&p21.)*2*&c23.) / ((&p11.-&p21.)*(&p22.-&p32.) - (&p21.-&p31.)*(&p12.-&p22.))

zero = 2 * ((&p21.-&p31.)*&c12. - (&p11.-&p21.)*&c23.) / ((&p11.-&p21.)*(&p22.-&p32.) - (&p21.-&p31.)*(&p12.-&p22.))

• Meine Zwischenschritte mit Stift und Papier  Mathematik Klasse 8?

• Was braucht ein SAS-Anwender?

1. Den wichtigsten Tipp, den Ralf Minkenberg uns/mir 2014 mit auf den Weg 
gegeben hat.

2. Die Fähigkeit des verstehenden Lesens.

• Der wichtigste Tipp

ods trace on;

proc discrim
…
run;

ods trace off;

Output Added:
-------------
Name:       LinearDiscFunc
Label:      Linear Discriminant Function
Template:   Stat.Discrim.LinearDiscFunc
Path:       Discrim.LinearDiscFunc
-------------

• Verstehendes Lesen

• Den SAS-Output lesen und verstehen.

• Das Log lesen und verstehen.

• <F1> vernünftig nutzen.

• Aus allen Drei das heraus ziehen können, um ohne manuelles Copy/Paste 
einen vernünftigen Output zu generieren.  

• Das Programm - Konstanten

ODS GRAPHICS / height=1000;

%LET sizemin=12px; /* Minimale Textgröße */

%LET ymin=-10; /* Y-Achse von */
%LET ymax=10; /* Y-Achse bis */
%LET xmin=-10; /* X-Achse von */
%LET xmax=10; /* X-Achse bis */
%LET step=5; /* Schrittweite */

%LET alpha=0.2; /* Größe der Streuungs-Ellipse */

%LET HK1=Prin1; /* Wahl der 1. Hauptachse */
%LET HK2=Prin2; /* Wahl der 2. Hauptachse */

%LET priors='1'=0.333 '2'=0.333 '3'=0.333;
/* A-priori-Wahrscheinlichkeiten der 3 Gruppen */

%LET wdh=30; /* Stichprobenumfang pro Test-Gruppe */
%LET diff=4; /* Auseinanderschieben der 3 Test-Gruppen */

• Das Programm - Gibt es Fehlklassifikationen?

/* Sind im Datensatz Fehlklassifikationen enthalten? */
%macro sizemax(data);

%Global sizemax;

%LET sizemax=16px;

proc sql noprint; 
select count(*) into :anz from &data. where size=2; 
quit;

/* Wenn keine Fehlklassifizierung (size=2) --> sizemax=sizemin */
%if &anz.=0 %then %LET sizemax=&sizemin.;

%mend;

• Das Programm - Testdaten erstellen

/* Ellipsoide 2-dimensionale Normalverteilung */
data tmp;
call streaminit(123);
do nr=1 to 3*&wdh.;
x=rand("Normal",0,2);
y=rand("Normal",0,3);
output;
end;
run;

/* Verschieben an 3 Punkte mit Abstand 2*Diff */
data three;
set tmp;
if nr<=&wdh. then do; grp=1; x=x-&diff.; end;
if (&wdh.+1)<=nr<=(2*&wdh.) then do; grp=2; t=x; x=y+&diff.; y=t-&diff.; end;
if (2*&wdh.+1)<=nr then do; grp=3; t=x; x=y+&diff.; y=t+&diff.; end;
drop t;
run;

• Das Programm - Testdaten ansehen

/* Anschauen */
proc sgplot data=three;

styleattrs datacolors=(blue red green) datacontrastcolors=(blue red green);

scatter x=x y=y / group=grp markerattrs=(symbol=circlefilled);

ellipse x=x y=y / group=grp alpha=&alpha.;

xaxis display=(nolabel) values=(&xmin. to &xmax. by &step.);
yaxis display=(nolabel) values=(&ymin. to &ymax. by &step.);

label grp='00'x;

run;

• Das Programm - Hauptkomponenten-Transformation

/* Hauptachsentransformation */
proc princomp data=three out=pcathree cov;
id nr grp;
var x y;
run;

/* Anschauen */
proc sgplot data=pcathree;
styleattrs datacolors=(blue red green) datacontrastcolors=(blue red green);

scatter x=&HK1. y=&HK2. / group=grp markerattrs=(symbol=circlefilled);

ellipse x=&HK1. y=&HK2. / group=grp alpha=&alpha.;

xaxis display=(nolabel) values=(&xmin. to &xmax. by &step.);
yaxis display=(nolabel) values=(&ymin. to &ymax. by &step.);
label grp='00'x;

run;

• Das Programm - Diskriminanzanalyse

/* Diskriminanzanalyse mit den Originaldaten */
proc discrim data=three pool=yes method=normal listerr;
class grp;
priors &priors.;
var x y;
run;

/* Diskriminanzanalyse mit den transformierten Daten */
proc discrim data=pcathree method=normal pool=yes listerr outstat=stats out=outerror;
class grp;
priors &priors.;
var &HK1. &HK2.;
run;

Oder:

ods output LinearDiscFunc=stats;
proc discrim data=irispca method=normal pool=yes listerr out=outerror;
class grp;
priors &priors.;
var prin:;
run;

• Das Programm - Diskriminanzanalyse

• Das Programm - Ergebnistabellen

• Das Programm - Diskriminanzanalyse

• Das Programm - Ergebnistabellen

• Das Programm - Parameter ermitteln

/* Variable mit den falsch klassifizierten Objekten erstellen */
proc sql noprint;
select nr into :wrong separated by ' ' from
(select 0 as nr from outerror union select nr from outerror where grp <>_INTO_);
quit;

%put &wrong;
0 78

• Das Programm - Parameter ermitteln

/* Diskriminanzfuntionen laden */
proc sql noprint;
select &HK1.,&HK2. into :p11,:p12 from stats where grp=1 and _name_='_LINEAR_';
select &HK1.,&HK2. into :p21,:p22 from stats where grp=2 and _name_='_LINEAR_';
select &HK1.,&HK2. into :p31,:p32 from stats where grp=3 and _name_='_LINEAR_';
select &HK1. into :p01 from stats where grp=1 and _name_='_CONST_';
select &HK1. into :p02 from stats where grp=2 and _name_='_CONST_';
select &HK1. into :p03 from stats where grp=3 and _name_='_CONST_';
quit;

Oder:

proc sql noprint;
select _1,_2,_3 into :p11,:p21,:p31 from stats where Variable="&HK1.";
select _1,_2,_3 into :p12,:p22,:p32 from stats where Variable="&HK2.";
select _1,_2,_3 into :p01,:p02,:p03 from stats where Variable='Constant';
quit;

%put &p01. &p02. &p03. &p11. &p12. &p21. &p22. &p31. &p32.;
-4.53531  -4.0109 -3.13992 -1.15061 -0.68844 1.139269 -0.22042 0.011341  0.90886

• Das Programm - Trennlinien errechnen  Versuch und Irrtum 

/* Berechnung der Konstanten aus den Diskriminanzfuntionen */
%LET c12 = %sysevalf((&p01. - &p02.)/2);
%LET c13 = %sysevalf((&p01. - &p03.)/2);
%LET c23 = %sysevalf((&p02. - &p03.)/2);

/* Berechnung der Trennlinien von jeweils 2 Gruppen */
data trennlinie12;
y12=&ymin.; x12=-(2*&c12.+(&p12.-&p22.)*y12)/(&p11.-&p21.); output;
y12=&ymax.;  x12=-(2*&c12.+(&p12.-&p22.)*y12)/(&p11.-&p21.); output;
run;

data trennlinie23;
y23=&ymin.; x23=-(2*&c23.+(&p22.-&p32.)*y23)/(&p21.-&p31.); output;
y23=&ymax.;  x23=-(2*&c23.+(&p22.-&p32.)*y23)/(&p21.-&p31.); output;
run;

data trennlinie13;
y13=&ymin.; x13=-(2*&c13.+(&p12.-&p32.)*y13)/(&p11.-&p31.); output;
y13=&ymax.;  x13=-(2*&c13.+(&p12.-&p32.)*y13)/(&p11.-&p31.); output;
run;

• Das Programm - Trennlinien errechnen

• Das Programm - Trennlinien errechnen

/* Merken der Fehlklassifikationen */
proc sql noprint;

select nr into :wrong1 separated by ' ' from
(select 0 as nr from outerror
union select nr from outerror where grp <>_INTO_ and _INTO_=1);

select nr into :wrong2 separated by ' ' from
(select 0 as nr from outerror
union select nr from outerror where grp <>_INTO_ and _INTO_=2);

select nr into :wrong3 separated by ' ' from
(select 0 as nr from outerror
union select nr from outerror where grp <>_INTO_ and _INTO_=3);

quit;

%put &wrong1. &wrong2. &wrong3. ;
0                 0 78 0

• Das Programm - Trennlinien von 2 Gruppen errechnen

/* Gruppen 2 und 3 */
data threelinie23;
merge pcathree (where=(grp<>1)) trennlinie23;
size=1;
if NR in (&wrong2. &wrong3.) then size=2;
run;

%sizemax(threelinie23);

proc sgplot data=threelinie23;
styleattrs datacolors=(red green) datacontrastcolors=(red green);
inset "Große Zahlen = falsch klassifizierte Objekte";
text x=&HK1. y=&HK2. text=nr / 

group=grp sizeresponse=size sizemin=&sizemin. sizemax=&sizemax.;
series x=x23 y=y23 / lineattrs=(color=black thickness=2);
ellipse x=&HK1. y=&HK2. / group=grp alpha=&alpha.;
xaxis display=(nolabel) values=(&xmin. to &xmax. by &step.);
yaxis display=(nolabel) values=(&ymin. to &ymax. by &step.);
label grp='00'x;
run;

• Das Programm - Trennlinien von 2 Gruppen errechnen

• Das Programm - die wichtigste Zeile  Schnittpunkt

%LET zero = %sysevalf( 2 * 
(  (&p21.-&p31.)*&c12. - (&p11.-&p21.)*&c23.  )  / 
(  (&p11.-&p21.)*(&p22.-&p32.) - (&p21.-&p31.)*(&p12.-&p22.)  )
);

Erinnerung an Studium: alle Diskriminanzfunktionen schneiden sich im selben Punkt

• Das Programm - verkürzte Trennlinien

/* Erstellung der verkürzten Linien */

data trennlinie12; /* green */
y12=&zero.; x12=-(2*&c12.+(&p12.-&p22.)*y12)/(&p11.-&p21.); output;
y12=&ymin.;  x12=-(2*&c12.+(&p12.-&p22.)*y12)/(&p11.-&p21.); output;
/* y12=&ymax.;  x12=-(2*&c12.+(&p12.-&p22.)*y12)/(&p11.-&p21.); output; */
run;

data trennlinie23; /* blue */
y23=&zero.;  x23=-(2*&c23.+(&p22.-&p32.)*y23)/(&p21.-&p31.); output;
y23=&ymax.; x23=-(2*&c23.+(&p22.-&p32.)*y23)/(&p21.-&p31.); output;
/* y23=&ymin.; x23=-(2*&c23.+(&p22.-&p32.)*y23)/(&p21.-&p31.); output; */

run;

data trennlinie13; /* red */
y13=&zero.; x13=-(2*&c13.+(&p12.-&p32.)*y13)/(&p11.-&p31.); output;
y13=&ymax.;  x13=-(2*&c13.+(&p12.-&p32.)*y13)/(&p11.-&p31.); output;
/* y13=&ymin.;  x13=-(2*&c13.+(&p12.-&p32.)*y13)/(&p11.-&p31.); output; */
run;

• Das Programm - alles jetzt in einen Datensatz

/* Gemeinsamer Datensatz */
data threelinie;
merge pcathree trennlinie12 trennlinie13 trennlinie23;
size=1;
if NR in (&wrong.) then size=2;
run;

/* Prüfen, ob Fehlklassifikationen */
%sizemax(threelinie);

• Das Programm - alles jetzt in einen Datensatz

/* Anschauen mit Trennlinien */
proc sgplot data=threelinie;

styleattrs datacolors=(blue red green) datacontrastcolors=(blue red green);

inset "Große Zahlen = falsch klassifizierte Objekte";

text x=&HK1. y=&HK2. text=nr / 
group=grp sizeresponse=size sizemin=&sizemin. sizemax=&sizemax.;

series x=x12 y=y12 / lineattrs=(color=green thickness=2); 
series x=x13 y=y13 / lineattrs=(color=red thickness=2);
series x=x23 y=y23 / lineattrs=(color=blue thickness=2);

ellipse x=&HK1. y=&HK2. / group=grp alpha=&alpha.;

xaxis display=(nolabel) values=(&xmin. to &xmax. by &step.);
yaxis display=(nolabel) values=(&ymin. to &ymax. by &step.);
label grp='00'x;

run;

Blau = 0.33
Rot  = 0.33
Grün = 0.33

Blau = 0.60
Rot  = 0.20
Grün = 0.20

Blau = 0.60
Rot  = 0.10
Grün = 0.30

Es sieht nicht 
immer schön 
aus, hier am 
Beispiel der 
IRIS-Daten

Download Demo:

https://sf.julius-kuehn.de/sas

https://sf.julius-kuehn.de/sas

